
Abstract
We motivate and discuss a novel functional programming construct
that allows convenient modular run-time nonstandard interpretation
via reflection on closure environments. Thismap-closure con-
struct encompasses both the ability to examine the contents of a
closure environment and to construct a new closure with a modi-
fied environment. From the user’s perspective,map-closure is a
powerful and useful construct that supports such tasks as tracing,
security logging, sandboxing, error checking, profiling, code in-
strumentation and metering, run-time code patching, and resource
monitoring. From the implementor’s perspective,map-closure
is analogous tocall/cc. Just ascall/cc is a non-referentially-
transparent mechanism that reifies the continuations that are only
implicit in programs written in direct style,map-closure is a non-
referentially-transparent mechanism that reifies the closure envi-
ronments that are only implicit in higher-order programs. Just as
CPS conversion is a non-local but purely syntactic transforma-
tion that can eliminate references tocall/cc, closure conversion
is a non-local but purely syntactic transformation that can elimi-
nate references tomap-closure. We show how the combination
of map-closure andcall/cc can be used to implementset! as
a procedure definition and a local macro transformation.

Categories and Subject Descriptors D.3.2 [Language Classifica-
tions]: Applicative (functional) languages; D.3.3 [Language con-
structs and features]: Procedures, functions, and subroutines

General Terms Design, Languages

Keywords Referential transparency, Lambda lifting

1. Motivation
Nonstandard interpretation is a powerful tool, with a wide variety
of important applications. Typical techniques for performing non-
standard interpretation are compile-time only, require modification
of global resources, or require rewriting of code to abstract over
portions subject to nonstandard semantics. This paper proposes a
construct to support modular run-time nonstandard interpretation.

For expository purposes, let us consider a very simple exam-
ple of nonstandard interpretation. Suppose one wished to add com-
plex numbers and complex arithmetic to a programming-language
implementation that supports only real arithmetic. One might rep-

resent the complex numbera + bi as an Argand pair〈a, b〉. Ex-
tending the programming language to support complex arithmetic
can be viewed as a nonstandard interpretation where real num-
bersr are lifted to complex numbers〈r, 0〉, and operations such
as+ : R × R → R are lifted to+ : C × C → C.

〈a1, b1〉 + 〈a2, b2〉 = 〈a1 + a2, b1 + b2〉

One can accomplish this in SCHEME by redefining the arith-
metic primitives, such as+, to operate on combinations of native
SCHEME reals and Argand pairs〈a, b〉 represented as SCHEME
pairs(a . b). For expository simplicity, we ignore the fact that
many of SCHEME’s numeric primitives can accept a variable num-
ber of arguments. We define a new procedurelift-+ which we
use to redefine+ at the top level.

(define (lift-+ +)
(lambda (x y)
(let ((x (if (pair? x) x (cons x 0)))

(y (if (pair? y) y (cons y 0))))
(cons (+ (car x) (car y))

(+ (cdr x) (cdr y))))))

(define + (lift-+ +))

This raises an important modularity issue. With the above defi-
nition, one can take a proceduref defined as

(define (f x) (+ x x))

and correctly evaluate(f ’(2 . 3)) to (4 . 6). One can even
take a procedureg defined as

(define g (let ((y 10)) (lambda (x) (+ x y))))

and correctly evaluate(g ’(1 . 2)) to (11 . 2). These exam-
ples work correctly irrespective of whetherf andg are defined be-
fore or after+ is redefined. In contrast, consider

(define h (let ((p +)) (lambda (x) (p x 5))))

The expression(h ’(1 . 2)) will evaluate correctly to(6 . 2)
only if h was defined after+ has been redefined. This is not the
only modularity issue raised by this common technique: for in-
stance, one might wish to confine the nonstandard interpretation
to a limited context; one might wish to perform different nonstan-
dard interpretations, either singly or cascaded; and one might wish
to avoid manipulation of global resources.

The remainder of this paper discusses a novel mechanism,
map-closure, which allows such nonstandard interpretation in
code written in a functional style, and which avoids these modular-
ity issues. As discussed in section 5,map-closure is a powerful
and useful construct that supports such tasks as tracing, security
logging, sandboxing, error checking, profiling, code instrumenta-
tion and metering, run-time patching, and resource monitoring.



2. The MAP-CLOSURE Construct
The inherent difficulty in the above example is that whilef andg
access the addition procedure using the global variable+, which
has been overloaded,h is closed over its own local variablep. In
most higher-order programming languages, closures are opaque:
there is no way to examine or modify closure environments. To ad-
dress the above modularity issue we introduce a reflection operator,
map-closure, which can examine the contents of a closure envi-
ronment and construct a new closure with a modified environment.

We adopt a simple model for closures in which they are rep-
resented as pairs containing an environment and a code pointer.
The environment in turn is represented as a mapping of free vari-
able names to values,{x1 7→ v1, . . . , xn 7→ vn}. The behavior of
map-closure is simple: it takes two procedure argumentsf andg,
and returns a procedureh whose code is identical to that ofg but
whose environment is{x1 7→ f(x1, v1), . . . , xn 7→ f(xn, vn)}.
We also allowmap-closure to apply to primitive procedures, tak-
ing their environments to be empty.

Note thatmap-closure is not referentially transparent. More-
over, the above formulation ofmap-closure is appropriate only
for dynamically-typed languages, since it uses a singlef to trans-
form all of the closed-over values ing and these may be of different
types. Incorporatingmap-closure into a statically typed language
would be an interesting topic for future research, as would be the
formulation of a referentially-transparent construct for first-class
nonstandard interpretation.

Given map-closure, we can define a procedure that substi-
tutesx for y in z.

(define (substitute x y z)
(cond
((equal? y z) x)
((pair? z)
(cons (substitute x y (car z))

(substitute x y (cdr z))))
((procedure? z)
(map-closure (lambda (n z) (substitute x y z))

z))
(else z)))

With this, we can overload the arithmetic operators inside closures.

(define (with-complex thunk)
((substitute (lift-+ +) + thunk)))

If our implementation represents constants as free variables,
we can alternatively lift not only the arithmetic operators but also
numeric values.

(define (lift-+ +)
(lambda (x y)
(cons (+ (car x) (car y))

(+ (cdr x) (cdr y)))))

(define (lift-r->c x)
(cond
((real? x) (cons x 0))
((pair? x)
(cons (lift-r->c (car x))

(lift-r->c (cdr x))))
((procedure? x)
(map-closure (lambda (n x) (lift-r->c x)) x))
(else x)))

(define (with-complex thunk)
((substitute (lift-+ +) + (lift-r->c thunk))))

Note that this removes the dispatches at every call to+ in the pre-
vious version, a sort of dynamic typing, and instead lifts all real
values into complex numbers once, at reflection time: a sort of run-
time static typing. However, just ascall/cc [1] complicates cer-
tain compiler optimizations, the presence ofmap-closure forces
compilers wishing to perform certain types of optimization, such as
constant folding, to do additional static analysis.

It is important to emphasize the power thatmap-closure af-
fords as illustrated bywith-complex in the above example. Sup-
pose a user obtains some library code that uses addition. As writ-
ten, the library code only performs real arithmetic since it uses the
built in addition primitive. The user can extend that library code
to apply to complex arithmeticwithout access to or modification
of the source code. The user does not need to wrap either the def-
inition of addition nor the procedures in the library with calls to
with-complex. All the user need do is wrap call sites to the li-
brary with calls towith-complex. In fact, the user need not do
even this: wrapping the top-level main program with a single call
to with-complex suffices.

The key contribution here is thatmap-closure is a general-
purpose programming construct that allows nonstandard interpre-
tation of library code without programmer access to and modifica-
tion of said library code. Moreover, if primitives are not (prema-
turely) inlined,map-closure allows such nonstandard interpreta-
tion without compiler access to or modification of source code.

For purposes of exposition we have motivatedmap-closure
using the well-known concept of complex arithmetic. Our interest
in this mechanism arose in a slightly different context: an attempt to
integrate a first-class derivative-taking operator into SCHEME using
the method of forward-mode automatic differentiation [2], which
performs arithmetic on dual numbersx + x′ε in a fashion roughly
analogous to that used for complex numbers [3].

There is an interesting analogy betweenmap-closure and
call/cc. Just ascall/cc reifies the continuations that are only
implicit in a program written in direct style,map-closure rei-
fies the closure environments that are only implicit in a higher-
order program. It is well known that CPS conversion [4, 5]
eliminatescall/cc. Similarly, closure conversion [6] eliminates
map-closure. We illustrate this using a simple dialect of SCHEME
defined in the next section. This implementation is available at
http://www.bcl.hamilton.ie/~qobi/map-closure/.

With both of the above implementations ofwith-complex the
scope of the lifting is limited to the dynamic extent of the invocation
of the thunk argument. However, it is tempting to try to implement
complex arithmetic in a form where the lifting is permanent. This
can be accomplished by invoking(call/cc with-complex), as-
suming that the implementation does not impose any arbitrary re-
strictions on the arguments tomap-closure. Note that this ap-
plies to both the past and the future. It modifies all bindings, both
past and future, via recursive descent into the continuation. The
mechanics and consequences of allowingmap-closure to be ap-
plied to continuations are discussed below. The next section de-
scribes a simple functional language which facilitates exploration
of the constructs under consideration. We then review the pro-
cess of closure conversion, show how closure conversion elimi-
natesmap-closure, and discuss howmap-closure in combina-
tion with call/cc allows set! to be implemented as a defined
procedure and a local macro transformation.

3. A Functional Subset of SCHEME

We formulate these ideas using a simple functional language that
resembles SCHEME [7], differing in the following respects:
• The only data types supported are Booleans, reals, pairs, and

procedures.



• Only a subset of the builtin SCHEME procedures and syntax are
supported.

• Rest arguments are not supported.

• The constructscons andlist are macros:
(cons e1 e2)  ((cons-procedure e1) e2)

(list)  ’()
(list e1 e2 . . .)  (cons e1 (list e2 . . .))

• Procedure parametersp can be variables,’() to indicate an
argument that is ignored, or(cons p1 p2) to indicate the
appropriate destructuring.

• All procedures take exactly one argument and return exactly
one result. This is accomplished in part by the basis, in part by
the following transformations:

(e1)  (e1 ’())
(e1 e2 e3 e4 . . .)  (e1 (cons* e2 e3 e4 . . .))

(lambda () e)  (lambda ((cons*)) e)
(lambda (p1 p2 p3 . . .) e)
 (lambda ((cons* p1 p2 p3 . . .)) e)

together with acons* macro

(cons*)  ’()
(cons* e1)  e1

(cons* e1 e2 e3 . . .)  (cons e1 (cons* e2 e3 . . .))

and by allowinglist andcons* as parameters.

The above, together with the standard SCHEME macro expansions,
a macro forif
(if e1 e2 e3)  
((if-procedure e1 (lambda () e2) (lambda () e3)))

and a transformation ofletrec into the Y-combinator suffice to
transform any program into the following core language:

e ::= (quote v) | x | (e1 e2) | (lambda (x) e)

4. Closure Conversion
The essence of closure conversion is to reify environments that con-
tain the values of free variables in procedures by replacing pro-
cedures with pairs of environments and a transformed procedure.
These transformed procedures have no free variables, and instead
access the values of free variables from the reified environment
passed as an argument. This can be implemented as a purely syn-
tactic source-to-source transformation, as shown in figure 1.

We omit a number of bookkeeping details tangential to the is-
sues we wish to explore. However, one bookkeeping issue relevant
to our purpose does arise. We would like our new reflective mech-
anism to be invariant to choice of variable names. We therefore
introduce a new data type,name, to key environments. The inter-
face for names consists of the proceduresname? andname=?, and
the syntax(name x) which returns a unique name associated with
the (potentially alpha-renamed) variablex.

Given this transformation,map-closure can be transformed to

(lambda (c (cons (cons f fc) (cons g gc)))
(cons g

(map (lambda (gn gv)
(cons gn (f fc gn gv)))
gc)))

The techniques described in this section and shown in figure 1 suf-
fice to implement the examples of section 2. While the simple im-
plementation in figure 1 represents reified closure environments as
alists and transformed procedures as pairs,map-closure does not
expose this structure. An alternate implementation could thus use

an alternate representation with suitable replacements forlookup,
map, and the locations in the transformation where closures are con-
structed. Such an implementation might represent names as offsets
into environment tuples.

5. The Utility of MAP-CLOSURE
Both alone and in combination withcall/cc, map-closure is
a powerful and general-purpose construct that can solve important
software-engineering tasks. It is a portable mechanism for perform-
ing run-time dynamic nonstandard interpretation, a technique of in-
creasing importance that arises in many guises ranging from secu-
rity and debugging to web applications (mechanisms likeAJAX that
overload I/O operations to useHTTP/HTML ). Consider the follow-
ing examples as an indication of its myriad potential uses.

Programmers often desire the ability to examine an execution
trace. Figure 2 contains atrace procedure that tracesall procedure
entrances and exits during the invocation ofthunk. Such a facility
can easily be adapted to perform security logging.

Virtual machines are often able to execute code in a sandbox so
as to constrain the allowed actions and arguments. Figure 2 contains
a sandbox procedure that invokesthunk in a context whereall
procedure invocations must satisfy theallowed? predicate or else
theraise-exception procedure is called. Such a facility is useful
both for security and error checking.

Many programming-language implementations contain a facil-
ity to profile code. Figure 2 contains aprofile procedure that
constructs a table of the invocation counts ofall procedures in-
voked during the invocation ofthunk. Such a facility can easily
be adapted to instrument and meter code in other ways.

One of the hallmarks of classical LISP implementations is the
ability to patch code in a running system by changing the function
bindings of symbols. The designers of COMMON L ISP were aware
that this mechanism could not be used to patch code referenced
in closure slots. They addressed this with a kludge: treating a
funcall to a symbol as afuncall to its function binding. Figure 2
contains a more principled approach to this problem. The procedure
patch replacesall live instances ofold with new.

Finally, many programming-language implementations contain
a facility to determine the amount of live storage. Figure 2 contains
aroom procedure that returns a list of the number of live pairs and
the number of live closure slots.

Facilities such as the above are normally implemented as sys-
tem internals. Figure 2 shows that many such facilities can be im-
plemented as user code withmap-closure.1

6. MAP-CLOSURE + CALL/CC = SET!

It is interesting to consider the application ofmap-closure to
a continuation made explicit bycall/cc. The source-to-source
transformation of closure conversion described in section 4 does
not allow this, because it does not closure-convert continuations.
However, we could convert the program to continuation-passing
style (CPS) first and then apply closure conversion, thus exposing
all continuations to closure conversion as ordinary procedures. Fig-
ure 3 describes this process. The transformations shown are stan-
dard, with one exception: themap-closure procedure itself needs
to be handled specially, as (prior to closure conversion) it cannot
be expressed as a user-defined procedure, and must be treated as
a primitive. However, it is unique among primitives in that it in-
vokes a procedural argument. Since this procedural argument will

1 The examples in figure 2 use a number of constructs not (yet) provided by
our prototype implementation, namely (implicit)begin, write, newline,
eq?, memq, and namedlet. It is straightforward to add such constructs.



(C c (quote v))  (quote v)
(C c (name x))  (name x)
(C c x)  (lookup (name x) c)
(C c (lambda (x) e))  (cons (lambda (c1 x) (let ((c2 (cons (cons (name x) x) c1))) (C c2 e)))

(list (cons (name x1) (C c x1)) . . .))
wherex1 . . . are free in(lambda (x) e)

(C c (e1 e2))  (let ((x (C c e1))) ((car x) (cdr x) (C c e2)))
e0  (let ((x1 (cons (lambda (c x) (x1 x)) ’())) . . .

(cons-procedure
(cons (lambda (c1 x1) (cons (lambda (c2 x2) (cons x1 x2)) ’())) ’()))
(map-closure
(cons (lambda (c (cons (cons f fc) (cons g gc)))

(cons g (map (lambda (gn gv) (cons gn (f fc gn gv))) gc)))
’()))

(pair? (cons (lambda (c x) (and (pair? x) (not (procedure? (car x)))))
’()))

(procedure? (cons (lambda (c x) (and (pair? x) (procedure? (car x))))
’())))

(let ((x (list (cons (name x1) x1) . . .
(cons (name cons-procedure) cons-procedure)
(cons (name map-closure) map-closure)
(cons (name pair?) pair?)
(cons (name procedure?) procedure?))))

(C x e0)))
wherex1 . . . are free ine0 exceptcons-procedure, map-closure, pair?, andprocedure?. This
assumes thatx1 . . . are bound to procedures that do not internally invoke procedural arguments.

Figure 1. Closure-conversion algorithm that applies to the top-level expressione0.

be in CPS after conversion, the CPS version ofmap-closure must
invoke this argument with an appropriate continuation.

The combination ofmap-closure andcall/cc is very power-
ful: it can be used to implementset! as a procedure definition in a
language that does not have any built-in mutation operations. The
intuition behind this is thatset! changes the value of a variable for
the remainder of the computation;call/cc exposes the remain-
der of the computation as a reified continuation;map-closure can
make a new continuation just like the old one except that one par-
ticular variable has a new value; and thus invoking this new contin-
uation instead of the old continuation has precisely the same result
asset!. The simple definition shown in figure 4 accomplishes this
intuition. There is, however, one minor complication: the recursion
in set-in is necessary because the target variable might be present
in closures nested in the environments of other closures. As a result,
unlike most SCHEME implementations, whereset! takes constant
time, the implementation in figure 4 must traverse the continuation
to potentially perform substitution in multiple environments that
close over the mutated variable.

While the ability to implementset! as a procedure definition
combined with a local macro transformation is surprising and in-
triguing, it might be reasonable to consider this to be something
of a curiosity. The combination ofmap-closure andcall/cc is
extremely powerful, and thus potentially difficult to implement ef-
ficiently. Howevermap-closure in the absence ofcall/cc is still
a useful construct for implementing nonstandard interpretation, and
seems amenable to more efficient implementation. Thus, imple-
mentations supportingmap-closure might not in general be ex-
pected to allow its application to continuations. Of the examples in
figure 2, onlypatch androom rely on this ability.

7. Discussion
Functor-based module systems [8], overloading mechanisms such
as aspect-oriented programming [9], andmap-closure are related,
in that all three support nonstandard interpretation. The difference
is in the scope of that nonstandard interpretation. In a functor-based

module system, the scope is lexical. With overloading, the scope is
global. Withmap-closure, the scope is dynamic.

The dynamic scope ofmap-closure affords interesting control
over modularity. One can apply a nonstandard interpretation to only
part of a program. Or, different nonstandard interpretations to dif-
ferent parts of a program. Or, to different invocations of the same
part of a program. One can compose multiple nonstandard interpre-
tations, controlling the composition order when they do not com-
mute. For example, composing complex arithmetic with logging
arithmetic in different orders would allow one to control whether
one logged the calls to complex arithmetic or the calls to the oper-
ations used to implement complex arithmetic. Withmap-closure,
nonstandard interpretations become first-class entities.

If all aggregate data structures are Church-encoded as closures,
CPS conversion followed by closure conversion subsumes store
conversion: it explicitly threads a store, represented as an environ-
ment, through the program. However, compilers that perform both
CPS conversion and closure conversion generally do so in the op-
posite order. Just ascall/cc affords one the power of explicit con-
tinuations while allowing one to write in direct style,map-closure
affords one the power of explicit closure environments while allow-
ing one to write in higher-order style. The combination ofcall/cc
and map-closure affords the power of explicit store threading
while allowing one to write in a direct higher-order style.

In the implementation ofset! in figure 4, the original continu-
ation is not mutated but discarded. Instead of discarding this orig-
inal continuation, it can be preserved and invoked later in order
to implement such control structures asfluid-let [5] and amb
[10] with associated side effects that are undone upon backtrack-
ing [11]. Side effects that can be undone can be used to implement
PROLOG-style logic variables and unification [12]. All this can be
implemented as defined procedures and local macro transforma-
tions in a language that has no explicit mutation operations, but
that supportscall/cc andmap-closure, allowingmap-closure
to apply to continuations.

Like other powerful constructs,map-closure may seem dif-
ficult to implement efficiently. However, the same was said of
constructs like recursion, dynamic typing, garbage collection, and



(define (trace thunk)
((let wrap ((x thunk))
(cond ((pair? x) (cons (wrap (car x)) (wrap (cdr x))))

((procedure? x)
(lambda (arguments)
(write (list +1 procedure arguments))
(newline)
(let ((result ((map-closure (lambda (n x) (wrap x)) x) arguments)))
(write (list -1 procedure result))
(newline)
result)))

(else x)))))

(define (sandbox allowed? raise-exception thunk)
((let wrap ((x thunk))
(cond ((pair? x) (cons (wrap (car x)) (wrap (cdr x))))

((procedure? x)
(lambda (arguments)
(if (allowed? x arguments) ((map-closure (lambda (n x) (wrap x)) x) arguments) (raise-exception))))

(else x)))))

(define (profile thunk)
(let* ((table ’())

(result ((let wrap ((x thunk))
(cond ((pair? x) (cons (wrap (car x)) (wrap (cdr x))))

((procedure? x)
(lambda (arguments)
(set! table (let increment ((table table))

(cond ((null? table) (list (cons x 1)))
((eq? (car (car table)) x)
(cons (cons (car (car table)) (+ (cdr (car table)) 1)) (cdr table)))
(else (cons (car table) (increment (cdr table)))))))

((map-closure (lambda (n x) (wrap x)) x) arguments)))
(else x))))))

(write table)
(newline)
result))

(define (patch old new)
(call/cc (lambda (c)

((let wrap ((x c))
(cond ((eq? x old) new)

((pair? x) (cons (wrap (car x)) (wrap (cdr x))))
((procedure? x) (map-closure (lambda (n x) (wrap x)) x))
(else x)))

#f))))

(define (room)
(let ((pairs 0) (slots 0) (objects ’()))
(call/cc (lambda (c)

(let walk ((x c))
(cond ((memq x objects) #f)

(else (set! objects (cons x objects))
(cond ((pair? x) (set! pairs (+ pairs 1)) (walk (car x)) (walk (cdr x)))

((procedure? x) (map-closure (lambda (n x) (set! slots (+ slots 1)) (walk x)) x))))))))
(list pairs slots)))

Figure 2. Typical LISP and SCHEME system functionality implemented as user code withmap-closure.

(C c (quote v))  (c (quote v))
(C c (name x))  (c (name x))
(C c x)  (c x)
(C c (lambda (x) e))  (c (lambda (c1 x) (C c1 e)))
(C c (e1 e2))  (C (lambda (x1) (C (lambda (x2) (x1 c x2)) e1)) e2)
e0  (let ((x1 (lambda (c x) (c (x1 x)))) . . .

(call/cc (lambda (c x1) (x1 c (lambda (c2 x2) (c x2)))))
(cons-procedure (lambda (c1 x1) (c1 (lambda (c2 x2) (c2 (cons x1 x2))))))
(map-closure (lambda (c (cons f g))

(c (map-closure (lambda (x) (f (lambda (x) x) x)) g)))))
(C (lambda (x) x) e0))

wherex1 . . . are free ine0 exceptcall/cc, cons-procedure, andmap-closure. This assumes
thatx1 . . . are bound to procedures that do not internally invoke procedural arguments.

Figure 3. CPS-conversion algorithm that applies to the top-level expressione0.

(define (set-in n v c)
(cond ((procedure? c) (map-closure (lambda (n1 v1) (if (name=? n n1) v (set-in n v v1))) c))

((pair? c) (cons (set-in n v (car c)) (set-in n v (cdr c))))
(else c)))

(define (set n v) (call/cc (lambda (c) ((set-in n v c) #f))))

(define-syntax set! (syntax-rules () ((set! x e) (set (name x) e))))

Figure 4. An implementation ofset! usingmap-closure andcall/cc.



call/cc when first introduced. Of particular concern is that it
may appear thatmap-closure precludes compiler optimizations
such as inlining, especially of primitive procedures. Well known
techniques (e.g., declarations, module systems, and flow analy-
sis) allow SCHEME compilers to perform inlining despite the fact
that the language allows redefinition of (primitive) procedures.
These techniques can be extended and applied to allow inlining
in the presence ofmap-closure. Even without such techniques,
map-closure does not preclude inlining: a compiler can generate
whatever code it wishes, so long as the run-time system can re-
construct the closure-slot information thatmap-closure passes to
its first argument, and any information needed to construct the re-
sult closure. Each invocation ofmap-closure might even perform
run-time compilation, including optimizations such as inlining.

The history of programming-language research is replete with
examples of powerful constructs that were initially eschewed for
performance reasons but later became widely adopted as their
power was appreciated and performance issues were addressed.
We hope that this will also be the case formap-closure.

Note that, by design,map-closure does not expose the internal
representation of closures and environments to the user. This design
also preserves hygiene: the lexical hierarchy of variable scoping.
Sincemap-closure does not allow one to add, remove, or rename
variables, it is not possible to create unbound variable references
or change the lexical scoping of variables through shadowing or
unshadowing at run time.

An alternate, more traditional way to provide the functionality
of map-closure would be to provide an interface to access the
environment and code components of closures and construct new
closures out of such environment and code components, along with
an interface to access environment components and construct new
environments. However, such an alternate interface would expose
the internal representation of closures and environments to the user,
perhaps via interfaces and data types that differ in detail between
implementations, and might well break hygiene. On the other hand,
map-closure exposes only one new data type: names as passed
as the first argument to the first argument ofmap-closure. The
values passed as the second argument to the first argument of
map-closure and the values returned by the first argument of
map-closure are ordinary SCHEME values.

Also note that names are opaque. They are created by new syn-
tax to allow implementations to treat them as variables in every
sense. They can only be compared via identity, so an implemen-
tation is free to represent names in the same way as variable ad-
dresses: stack offsets, absolute global addresses, etc. In fact, just as
implementations can have different representations of variable ad-
dresses for variables of different types and lifetimes, implementa-
tions can have similarly different representations of names. More-
over names can be avoided entirely by using a weaker variant of
map-closure that only exposes closure-slot values. Such a weaker
variant suffices for many applications, including all examples here
except for the implementation ofset!.

Closure conversion is not the only implementation strategy for
map-closure. For instance, a native implementation could oper-
ate directly on higher-order code. Such an implementation would
only need a mechanism for accessing slots of existing closures and
creating closures with specified values for their slots. These mech-
anisms already exist in any implementation of a higher-order lan-
guage, and must simply be repackaged as part of the implementa-
tion of amap-closure primitive. Furthermore, native implemen-
tations ofmap-closure are possible in systems that use alternate
closure representations, such as linked or display closures, unlike
the flat-closure representation used here. While theimplementation
of map-closure for different representations of closures and en-
vironments would be different, programs thatusemap-closure

would be portable across all such implementations. This is not the
case with the aforementioned alternate interface.

Nonstandard interpretation is ubiquitous in programming lan-
guage theory, manifesting itself in many contexts. It could be rea-
sonably suggested that the lack of a simple way to easily perform
a nonstandard interpretation may have held back the application of
this powerful idea, and resulted in a great deal of implementation
effort building systems that each perform some specific nonstan-
dard interpretation. For this reasonmap-closure, or some other
construct that provides first-class dynamic nonstandard interpreta-
tion, may prove a surprisingly handy tool. In fact, the authors have
already found it quite useful in the implementation of automatic
differentiation in a functional programming language.

